

## Our Adverse Event Review Reports Generated All in ODS Report Writing Interface

Sijian Zhang, INTERMACS, University of Alabama at Birmingham

### ABSTRACT

It is a common practice that third party software is used for generating complex reports. Usually, the data or some report components are prepared in SAS®, and then other reporting jobs are outsourced to other software, such as MS Word. This situation has been changed since SAS 9.2. Taking one of our routine reports as an example, this paper will illustrate what the ODS report writing interface can do, how syntaxes are applied, and how complicated reporting features are done in simple code. With this new tool, our complex reports can be done all in SAS in a smoother and more efficient way.

### KEYWORDS

Complex reports, ODS report writing interface, inline formatting.

### BACKGROUND

INTERMACS® (Interagency Registry for Mechanically Assisted Circulatory Support) is a national registry for patients who are receiving mechanical circulatory support device therapy to treat advanced heart failure. All the data is collected online through the United Network for Organ Sharing (UNOS) in Richmond, Virginia; and all the analyses and reports are processed in the INTERMACS® at the University of Alabama at Birmingham. We receive 37 SAS datasets from UNOS periodically, which include all kinds of information regarding mechanical heart transplantation.

The focus of this paper is the adverse event review reports, which are sent to the doctors periodically to verify if the stated events or causes are valid. The report is composed of two parts, on the top is the Patient Information Overview, and then followed by the Event Worksheets that correspond to the high-lighted events listed in the Patient Information Overview.

In a brief view of a sample report (see Appendix 1), you can see that it is not simple, not something which can be easily generated by SAS before versions 9.1. Besides the fancy layout of the report, here are two specifications that make it even harder.

- 1) The horizontal and vertical spaces of some items should be dynamically adjusted according to the lengths of the values.
- 2) The event worksheets should be ordered by the event date, which should be in the same order as the corresponding events listed in the Patient Information Overview sheet.

Before the ODS report writing interface is available in SAS, for situation like this, we used to apply SAS to prepare the report data, and then outsource the display jobs to MS Word, or export the data to MS ACCESS by which the reports were generated.

### INTRODUCTION

Before showing how the reports are generated, let's have a brief look at the two report writing tools, the ODS report writing interface and the inline formatting.

#### *ODS Report Writing Interface*

Every time the Interface is run, you will see a warning message in the log window (Figure 1). Since its status is "preproduction" in SAS 9.2, you cannot find relevant documentations in the SAS Help window. But you can still find a lot of information online. The references of this paper will give you a good starting point.

**WARNING: Data step interface is preproduction in this release.**

**Figure 1**

DATA \_NULL\_ has been used for report writing for a long time in SAS. The ODS report writing interface is a great leap further. Its power comes from the combination of DATA \_NULL\_ and ODS. It fully applies ODS features such as proportional fonts, colors, images, and so on; while at the same time it provides very flexible placement capabilities, and takes great advantage of the rich programming features that the data step offers, such as conditional logic, formatting capabilities, by-group processing, arrays, etc. The Interface is object-oriented, which provides you with many useful methods to control how you want to display your information so that even the most rigid reporting requirements can be met with ease.

Here is the basic programming structure of the Interface used in this paper.

```
ods listing close;
ods pdf notoc startpage=no style=printer_adj file="...\ReportName.pdf";

data _null_;
  set AE_Info;
  declare odsout adj();           ❶

  adj.table_start();              ❷
  adj.row_start();
  adj.format_cell(data: " Patient and Device Information",           ❸
    overrides: "just=1 font_size=14pt backgroundcolor=cxcccfff font_weight=bold");
  adj.row_end();
  adj.table_end();                ❹

  ...
run;
...
ods pdf close;
```

**❶ Declare an ODS object:**

Two ways:

- i) declare odsout object;  
object = \_new\_ object ( );
- ii) declare odsout object ( );

Here, “declare” (short form: dcl) is the key word for declaring an object; “odsout” is the key word (class name) for creating a class instance of ODS output object; and “object” is placeholder for any object variable name. The above two methods have the same effect. In the code above, “adj” is the object variable name. (The event review process was called “adjudication” initially.)

**❷ and ❹ Object methods used to set up a table:**

The syntax for an object to use methods:

object.method (<optional argument>, ..., <optional argument>);

In the above code, method table\_start( ) starts a table. It is always coupled with method table\_end( ), which ends the table; the methods row\_start( ) and row\_end( ) work in the same way; and the method format\_cell( ) works alone to define a cell.

**❸ Method arguments:**

They define what contents and styles are used to display by a method.

Here, the argument “data:” tells to show the text “ Patient and Device Information” in the cell; and the argument “override” tells that 4 style values will be reset in this cell instead of the default ones.

## Inline Formatting

The inline formatting syntax: `escape character {function-name <argument-1 <argument-2 ... <argument-n>>>}`

The inline formatting is a very useful ODS tool that applies formatting functions to define how the specific contents are displayed rather than using global or default styles. This tool is experimental in SAS 8.2, and is in production for all destinations in SAS 9.2. Here is an example in the report.

```
ods escapechar='^';  
title "^{style [just=left preimage='...\INTERMACS_logo_.bmp']}"  
      "^{nbspspace 35} ^{style [font_size=19pt font_weight=bold font_style=italic] Medical Event Review Worksheet}"  
      "^{newline} ^{style [just=right font_face=arial font_size=10pt] Event Date: before 4/1/2010}";
```

① To specify an escape character:

The syntax: `ods escapechar='escape-character';`

An escape character should not occur for any other uses in the code. For the inline formatting, it indicates that an inline formatting function follows. The functions and the specified contents are wrapped in curly brackets. Here, '^' is specified as the escape character.

② Inline formatting functions in the above statement:

Style:       Modifies the style of the current contents. "preimage=" argument imports an image at the beginning of the title.  
Nbspace:     Insert blank spaces.  
Newline:     Start a new line.

Here is how the above statement displays the title in the report:



Figure 2

You can apply the inline formatting to any contents (inserted text or variable value) that you want to display in a report. The coding is simple, and it will make your report look great.

## APPLICATION IMPLEMENTATION

The code to generate the report can be grouped into two parts: data preparation and report writing. The focus in this paper is the second part. However, to better understand the report writing, a brief description of the first part is helpful.

### Data Preparation

After the data extraction and manipulation from raw datasets, two sets of data are generated: patient overall information and the adverse event information. Within each folder (see Figure 3 and 4), the small datasets are subset from two large datasets by Event\_ID, which is postfix of each dataset name.

For example, the dataset "pt\_12.sas7bdat" lists all relevant clinical events after the operation with Event\_ID=12, ordered by the event date; while "ae\_12.sas7bdat" holds the information of all adverse events sorted by event date after the operation, which are selected for the doctors to review.




Figure 3




Figure 4

## Report Writing

In the report writing code, DATA \_NULL\_ is the only SAS step repeatedly used to generate all the pieces of the report. Instead of going through the statements line by line, several helpful programming points are illustrated in this section. If you want to find a specific syntax in detail, please view the references.

### 1. Dynamic Spacing within Reports

If a reporting program uses the fixed display setting, sometimes it is not easy to set the right cell sizes or the right spaces between cells on a report sheet. The trouble is how to display all the values, and in the meantime to have a fine layout of the report contents if there exit a few extreme long strings for certain variable fields. If you want the report to look good, you may have to truncate those long values or rephrase them. Using the Interface, given the report setting and the length of variable value, if it is necessary, a new row will be inserted in the cell automatically until the entire value is displayed; or the horizontal spaces will be adjusted automatically in an optimal way.

Comparing the following two figures (Figure 5 and 6), you will see the effect in horizontal and vertical dimensions.




Figure 5




Figure 6

In Figure 6, the spaces between items in the first row are adjusted automatically due to the long Race value; and a new row is added due to the long Patient Profile Status value. The following is the corresponding code.

```

%macro insert1(label,var);                                ❶
adj.format_cell(data: "^{style [foreground=cx0000ff]&label: }"||strip(&var), overrides: "just=1");
%mend insert1;

%macro insert2(label,var,unit);                            ❷
adj.format_cell(data: "^{style [foreground=cx0000ff]&label: }"||strip(&var)||" &unit", overrides:
"just=1");
%mend insert2;

%macro blkrow(height=1);                                  ❸
adj.row_start();
adj.format_cell(overrides: "cellheight=&height.mm");
adj.row_end();
%mend blkrow;
...
data _null_;
...
adj.table_start(overrides: "width=100pct borderwidth=0");
%blkrow();

adj.row_start();                                         ❹
%insert1(Hospital ID,Hospital_ID);
%insert1(Patient ID,Patient_ID);
%insert1(Event ID,Event_ID);
%insert1(Report ID,Patient_Report_ID);
%insert1(Gender,Gender);
%insert1(Race,Race);
%insert2(Age,Age,yr);
adj.row_end();
adj.table_end();
...
run;

```

- ❶ A macro to insert the label in blue and variable value in black into a cell.
- ❷ A macro to insert the label in blue and variable value in black plus a unit name into a cell.
- ❸ A macro to insert a blank row space with default height = 1mm.
- ❹ Putting all the cell into one row without setting “cellwidth” argument. By default, SAS will adjust them automatically. If you want to put them in fixed positions, you can do it by setting “cellwidth” values.

You do not see the specific coding for the adjusting effect, which is all done by SAS default. These little things may turn out to be a big programming deal in other software packages, while SAS automatically does these for you. What a relief!

## 2. Generating Headers in DATA \_NULL\_

In SAS, if a procedure generates the resulting tables or lists longer than one page, the column headers will be automatically added on the top of the new page, but this will not happen in DATA \_NULL\_ by default. However, we can still do it in the following code, which inserts the headers in Figure 7 on the top of every new page.

Event	Event Date	Submission Status	Last Saved
Demographics	10/12/2006 (Data entry start)	Complete	03/28/2007

Figure 7

```

%macro Packet(Event_ID,N,pageLines);
...
%macro Pt_Info;
data _null_;
  set Pt_Info end=eof;
  if _n_=1 then declare odsout adj();

  if _n_=1 | mod(_n_,&pageLines)=&pageLines-1 then do; ❶
adj.table_start(overrides: "borderwidth=2");
adj.row_start();
adj.format_cell(data: "Event", inhibit: "LR",
  overrides: "font_weight=bold vjust=m cellwidth=11.2cm");
adj.format_cell(data: "Event Date", inhibit: "LR",
  overrides: "font_weight=bold vjust=m cellwidth=3.7cm");
adj.format_cell(data: "Submission|Status", split: "|", inhibit: "LR", ❷
  overrides: "font_weight=bold cellwidth=1.8cm");
adj.format_cell(data: "Last|Saved", split: "|", inhibit: "LR",
  overrides: "font_weight=bold cellwidth=1.7cm");
adj.row_end();
adj.table_end();
end;
...
run;
%mend Pt_Info;
...
%mend Packet;

```

❶ The trick to put the headers on the top of each page is the MOD( ) function on \_N\_ in the IF statement. First, I counted the number of rows in one page in a testing run, then set the macro variable &pageLines to that number+1. So, if it is the top row of a page, the column headers will be inserted. The assumption for this approach to work is that the width of each cell is long enough to display all the characters of the variable. We do have very few cases that break this assumption. If it happens, we have to adjust the position to keep the headers on the top of page by setting the value of &pageLines individually.

❷ The argument "split" specifies the symbol to break the data argument contents into two rows; the argument "inhibit" suppresses the border lines of a cell, for example, inhibit: "LR" means that the left and right border lines of the cell are suppressed.

### 3. Special Symbols

Special symbols are often appeared in reports. In our reports, we used some square check boxes, such as in Figure 8.

Based on the provided information:

I. Occurrence of bleeding:

☐ I find no reason to question the existence of this bleeding episode as stated above.

☐ I do find reason to question the existence of this bleeding episode as stated above because:

Figure 8

Here is the code to use special symbols:

```

...
adj.row_start();
adj.format_cell(data: "^{style [font_size=13pt] ^{unicode 2610}} ^{style ❶
  [font_size=10pt]I find no reason to question the existence of this bleeding ....}",
  overrides: "just=1 cellheight=0.78cm");
adj.row_end();

```

```

adj.row_start();
adj.format_cell(data: "^{style [font_size=13pt] ^{unicode 2610}} ^{style
[font_size=10pt]I ^{style [font_size=9pt font_weight=bold
foreground=maroon] do} find reason to question the existence of this bleeding ...:}",
overrides: "just=1 cellheight=0.78cm");
adj.row_end();
...

```

- ❶ and ❷ We can use inline formatting UNICODE function to insert whatever special symbols available in Unicode list which has various symbols. Here, the Unicode 2610 stands for the square check box.

#### 4. Ordering Event Worksheets by Event Date

To order the worksheets by event date is an issue in our report programming. The order may not be a big deal for the situation in Appendix 1, in which only 3 event worksheets are included. However, if there are 30 events in one report, it will be messy for the reviewer to sort them out by the event date. The following is one way to handle this issue.

```

%macro Packet(Event_ID,N,pageLines);
...
%do i=1 %to &n;
data AE_&i;
set mem_ae.ae_&i;
if &i=_n_ then call symput('AE_Code',strip(ae));
if &i=_n_;
run;

ods pdf startpage=now;
ods pdf anchor="&i.ae";
%if &AE_Code=0.5 %then %Bleeding;
%else %if &AE_Code=1 %then %Device;
%else %if &AE_Code=2 %then %Infection;
%else %if &AE_Code=3 %then %Neuro;
%else %if &AE_Code=4 %then %Death;
%end;
...
%mend Packet;

```

- ❶ The loop will go through all the adverse events for each case in the dataset AE\_&Event\_ID that is sorted by event date (see Data Preparation section). The values for &N are from another dataset (Event\_ID\_N) that contains the frequencies of the events to review since the operation.
- ❷ To set one of the AE datasets in Figure 4.
- ❸ To assign a value to macro variable &AE\_Code.
- ❹ To select one observation from dataset AE\_&Event\_ID by &i, which is used in the next data step.
- ❺ To run the specific event worksheet generation macro according to &AE\_Code.

#### 5. Conditional Formatting

You may have noticed that in the “Event” column of the patient overall information sheet, some events stand out in a different format (blue and bold). It means that these events are selected for the current review.

<b>AE Bleeding</b> (Report ID: 18618)	12/17/2009	01/27/2010
Rehospitalization - Admission	12/17/2009	01/27/2010
Rehospitalization - Discharge	12/22/2009	01/27/2010
<b>AE Neuro Dysfunction</b> (Report ID: 18619)	12/25/2009	01/27/2010
Rehospitalization - Admission	12/25/2009	01/27/2010
<b>AE Bleeding</b> (Report ID: 18619)	12/29/2009	01/27/2010

Figure 9

Here is the code:

```
%macro Pt_Info;
data _null_;
...
    if fmt=1 then do; ❶
        adj.format_cell(data: "^{style [font_weight=bold]}||strip(Event)||"}||
        ' (Report ID: '||strip(patient_report_id)||')', inhibit: "LTR",
        overrides: "just=l vjust=t foreground=blue font_weight=light ❷
        cellwidth=11.2cm cellpadding=0 url="||strip(LinkTo));
    end;
...
run;
%mend Pt_Info;
```

- ❶ The variable `fmt` is in the `Pt_Event_ID` datasets in Figure 3, which indicates whether an event is going to be reviewed (`fmt = 1`: to be reviewed).
- ❷ By taking the advantage of data step, IF statement is used here to conditionally set the format for only the events to be reviewed. The style parameters in “data” and “overrides” arguments define the new format.

## CONCLUSION and COMMENT

The ODS report writing interface is an excellent tool to handle the complex reports. Comparing with other approaches, the Interface makes the report generation process smoother and more efficient. It gets all the programming jobs done in SAS, a one-stop solution. Based on my experiences, the Interface programming is easy to learn and very productive. The coding process might be tedious sometimes, and proper use of macros can reduce the repetitions.

I agree with what Daniel O'Connor stated in his paper, “DATA \_NULL\_ report writing has long been an integral part of the custom report writing offered by SAS®, but with this newly updated ODS Report Writing technology in SAS® 9.2, you will have the ability to produce reports that you have only dreamed about.”[2]

## REFERENCES

1. Appendix 2: Method Documentation, (a list of object method syntaxes with brief examples)  
support.sas.com/rnd/base/datastep/dsobject/Power\_to\_show\_documentation.pdf
2. Daniel O'Connor, The Power to Show: Ad Hoc Reporting, Custom Invoices, and Form Letters, Paper 313-2009, SAS Global Forum 2009, <http://support.sas.com/resources/papers/proceedings09/313-2009.pdf>

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Sijian Zhang, MS, MBA  
Database Administrator  
INTERMACS  
Division of Cardiothoracic Research  
Department of Surgery  
University of Alabama at Birmingham  
sijian@uab.edu



SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.



## Appendix 1.



### Patient Information Overview

(Event ID: 865)

Event	Event Date	Submission Status	Last Saved
Demographics	03/05/2008 (Data entry start)	Complete	12/31/2008
Pre-Implant	11/28/2007 (Decision Date)	Complete	01/22/2009
Quality of Life (Pre-Implant) - Not completed			01/22/2009
Trailmaking (Pre-Implant) - Not attempted			01/22/2009
Implant	11/30/2007	Complete	08/01/2008
1 Week Post-Implant	12/07/2007	Complete	08/01/2008
Implant Discharge	12/17/2007	Complete	08/04/2008
1 Month Post-Implant	12/30/2007 (Expected date)	Complete	08/01/2008
3 Month Follow-Up	02/12/2008	Complete	08/01/2008
Quality of Life (3-Month) - Not completed			08/01/2008
Trailmaking (3-Month) - Not attempted			08/01/2008
6 Month Follow-Up	05/06/2008	Complete	08/04/2008
Quality of Life (6-Month) - Completed			08/04/2008
Trailmaking (6-Month) - Not attempted			08/04/2008
AE Infection	06/23/2008		08/04/2008
Rehospitalization - Admission	06/23/2008		08/04/2008
Rehospitalization - Discharge	06/27/2008		08/04/2008
AE Bleeding	11/08/2008		04/07/2009
1 Year Follow-Up	11/25/2008	Complete	04/07/2009
Quality of Life (12-Month) - Not completed			04/07/2009
Trailmaking (12-Month) - Not attempted			04/07/2009
1.5 Year Follow-Up	05/28/2009	Complete	06/12/2009
Quality of Life (18-Month) - Not completed			06/12/2009
Trailmaking (18-Month) - Not attempted			06/12/2009
2 Year Follow-Up	12/17/2009	Complete	01/27/2010
Quality of Life (24-Month) - Not completed			01/27/2010
Trailmaking (24-Month) - Not attempted			01/27/2010
AE Bleeding (Report ID: 18618)	12/17/2009		01/27/2010
Rehospitalization - Admission	12/17/2009		01/27/2010
Rehospitalization - Discharge	12/22/2009		01/27/2010
AE Neuro Dysfunction (Report ID: 18619)	12/25/2009		01/27/2010
Rehospitalization - Admission	12/25/2009		01/27/2010
AE Bleeding (Report ID: 18619)	12/29/2009		01/27/2010
Rehospitalization - Discharge	12/31/2009		01/27/2010
Psychiatric Episode	01/27/2010 (Report Date)		
2.5 Year Follow-Up	05/30/2010 (Expected date)	Incomplete	

Generated on: 05/07/2010

-- 1 --



### Medical Event Review Worksheet

Event Date: before 4/1/2010

#### Patient and Device Information

Hospital ID: 81 Patient ID: 912 Event ID: 865 Report ID: 18618 Gender: M Race: White Age: 68 yr  
 Height: 175 cm Weight: 92 kg Previous Cardiac Operation: CABG  
 Primary Card Diag: Dilated Myopathy: Ischemic Admission Reason: Heart failure  
 Device Type: LVAD Device Brand: Heartmate XVE Implant Date: 11/27/2007  
 Patient Profile Status: 3 "Stable but inotrope dependent" describes a patient who is clinically stable on mild-moderate dose  
 Device Strategy: Destination Therapy (patient definitely not eligible for transplant)

#### Event Report: Bleeding

Date of Event: 12/17/2009 Months Post-Implant: 24.5755 Patient Location: In hospital  
 Conditions Resulting from Bleeding: Episode resulted in transfusion  
 Transfusion Date: 12/19/2009  
 Bleeding Units: 2-3 units  
 Bleeding Source/Cause/Location: Lower gastrointestinal  
 Drug Intervention: N  
 Causative Factors: Complexities of Medical Management  
 Bleeding Factor Conditions:  
 Lab Tests: INR: 1.3 Test Date: 12/24/2009  
 PTT: 28 Test Date: 12/24/2009  
 PH: Test Date:  
 Anti-coagulation Therapy: None  
 Anti-coagulation Therapy - Other:

#### Adjudication Results

##### Based on the provided information:

##### I. Occurrence of bleeding:

- ☐ I find no reason to question the existence of this bleeding episode as stated above.  
☐ I do find reason to question the existence of this bleeding episode as stated above because:

##### II. Causative or contributing factors to bleeding episode:

- ☐ I find no reason to question the factors related to this bleeding episode as stated above.  
☐ I do find reason to question the factors related to this bleeding episode as stated above because:

If you do not agree with the factor(s) above, please select causative or contributing factors to this bleeding (check all that apply):

- ☐ Poor compliance with monitoring anticoagulation therapy  
☐ Documented history of lesion or condition in the organ site  
 of bleeding that could potentiate a bleeding episode  
 Name of condition: \_\_\_\_\_  
☐ Elevated preoperative INR or platelet count less than 60,000  
☐ Management: Over anticoagulation  
☐ Complexities of Medical Management  
☐ Procedural related to Implant procedure  
☐ Procedural related to any re-operative procedure  
☐ Procedural related to any diagnostic procedure (e.g. bronchoscopy, endoscopy, or transesophageal echo)  
☐ Unknown

Signature of Committee Member: \_\_\_\_\_ Date: \_\_\_\_\_

Generated on: 05/07/2010

-- 2 --



## Medical Event Review Worksheet

Event Date: before 4/1/2010

### Patient and Device Information

Hospital ID: 81 Patient ID: 912 Event ID: 865 Report ID: 18619 Gender: M Race: White Age: 68 yr  
 Height: 175 cm Weight: 92 kg Previous Cardiac Operation: CABG  
 Primary Card Diag: Dilated Myopathy: Ischemic Admission Reason: Heart failure  
 Device Type: LVAD Device Brand: Heartmate XVE Implant Date: 11/27/2007  
 Patient Profile Status: 3 "Stable but inotrope dependent" describes a patient who is clinically stable on mild-moderate dose  
 Device Strategy: Destination Therapy (patient definitely not eligible for transplant)

### Event Report: Neurological Dysfunction

Date of Event: 12/25/2009 Months Post-Implant: 24.8383 Patient Location: Out of hospital  
 Neurological Category: Neurological Dysfunction - < 24 hours  
 Neurological Category - Other:  
 Causative Factor: Complexities of Medical Management  
 CNS Event: Intracranial Bleed  
 CNS Event - Other:  
 CNS Event Location: Left hemisphere  
 CNS Event Location - Other:  
 CNS Diagnosis Method: CT  
 CNS Diagnosis Method - Other:  
 Clinical Event: Stroke  
 Clinical Event - Stroke: Altered mental status  
 Clinical Event - Stroke - Other:  
 Surgical Intervention: N Drug Intervention: Y Drug Treatment: Thrombolytics  
 Contributes to Death: N

### Adjudication Results

#### Based on the provided information:

##### I. Occurrence of neurological dysfunction:

- ☐ I find no reason to question the existence of this neurological episode as stated above.  
☐ I do find reason to question the existence of this neurological episode as stated above because:

##### II. Causative or contributing factors to neurological dysfunction:

- ☐ I find no reason to question the factors related to this neurological episode as stated above.  
☐ I do find reason to question the factors related to this neurological episode as stated above because:

If you do not agree with the factor(s) above, please select causative or contributing factors to this neurological episode (check all that apply):

- ☐ Patient not taking anticoagulation medication properly ☐ If patient receiving heparin then, evidence of PTT below target range  
☐ If patient receiving warfarin then, evidence of INR above target range ☐ Complexities of Medical Management  
☐ If patient receiving warfarin then, evidence of INR below target range ☐ Device Related  
☐ If patient receiving heparin then, evidence of PTT above target range ☐ Unknown

Signation of Committee Member: \_\_\_\_\_ Date: \_\_\_\_\_

Generated on: 05/07/2010

3



## Medical Event Review Worksheet

Event Date: before 4/1/2010

### Patient and Device Information

Hospital ID: 81 Patient ID: 912 Event ID: 865 Report ID: 18619 Gender: M Race: White Age: 68 yr  
 Height: 175 cm Weight: 92 kg Previous Cardiac Operation: CABG  
 Primary Card Diag: Dilated Myopathy: Ischemic Admission Reason: Heart failure  
 Device Type: LVAD Device Brand: Heartmate XVE Implant Date: 11/27/2007  
 Patient Profile Status: 3 "Stable but inotrope dependent" describes a patient who is clinically stable on mild-moderate dose  
 Device Strategy: Destination Therapy (patient definitely not eligible for transplant)

### Event Report: Bleeding

Date of Event: 12/29/2009 Months Post-Implant: 24.9697 Patient Location: In hospital  
 Conditions Resulting from Bleeding: Episode resulted in transfusion  
 Transfusion Date: 12/29/2009  
 Bleeding Units: 2-3 units  
 Bleeding Source/Cause/Location: Lower gastrointestinal  
 Drug Intervention: Y  
 Causative Factors: Complexities of Medical Management  
 Bleeding Factor Conditions:  
 Lab Tests: INR: 1.3 Test Date: 12/29/2009  
 PTT: 40 Test Date: 12/29/2009  
 PH: Test Date:  
 Anti-coagulation Therapy: Warfarin, Aspirin  
 Anti-coagulation Therapy - Other:

### Adjudication Results

#### Based on the provided information:

##### I. Occurrence of bleeding:

- ☐ I find no reason to question the existence of this bleeding episode as stated above.  
☐ I do find reason to question the existence of this bleeding episode as stated above because:

##### II. Causative or contributing factors to bleeding episode:

- ☐ I find no reason to question the factors related to this bleeding episode as stated above.  
☐ I do find reason to question the factors related to this bleeding episode as stated above because:

If you do not agree with the factor(s) above, please select causative or contributing factors to this bleeding (check all that apply):

- ☐ Poor compliance with monitoring anticoagulation therapy ☐ Complexities of Medical Management  
☐ Documented history of lesion or condition in the organ site ☐ Procedural related to implant procedure  
of bleeding that could potentiate a bleeding episode ☐ Procedural related to any re-operative procedure  
Name of condition: \_\_\_\_\_ ☐ Procedural related to any diagnostic procedure (e.g.  
bronchoscopy, endoscopy, or transesophageal echo)  
☐ Elevated preoperative INR or platelet count less than 60,000  
☐ Management: Over anticoagulation ☐ Unknown

Signation of Committee Member: \_\_\_\_\_ Date: \_\_\_\_\_

Generated on: 05/07/2010

4